
Plug is:
	A specification for composing web applications with functions
	Connection adapters for different web servers in the Erlang VM

Documentation for Plug is available online.

  
    
  
  Installation


In order to use Plug, you need a webserver and its bindings for Plug.
There are two options at the moment:
	Use the Cowboy webserver (Erlang-based) by adding the plug_cowboy package to your mix.exs:
 def deps do
   [
     {:plug_cowboy, "~> 2.0"}
   ]
 end

	Use the Bandit webserver (Elixir-based) by adding the bandit package to your mix.exs:
 def deps do
   [
     {:bandit, "~> 1.0"}
   ]
 end



  
    
  
  Hello world: request/response


This is a minimal hello world example, using the Cowboy webserver:
Mix.install([:plug, :plug_cowboy])

defmodule MyPlug do
  import Plug.Conn

  def init(options) do
    # initialize options
    options
  end

  def call(conn, _opts) do
    conn
    |> put_resp_content_type("text/plain")
    |> send_resp(200, "Hello world")
  end
end

require Logger
webserver = {Plug.Cowboy, plug: MyPlug, scheme: :http, options: [port: 4000]}
{:ok, _} = Supervisor.start_link([webserver], strategy: :one_for_one)
Logger.info("Plug now running on localhost:4000")
Process.sleep(:infinity)
Save that snippet to a file and execute it as elixir hello_world.exs.
Access http://localhost:4000/ and you should be greeted!

  
    
  
  Hello world: websockets


Plug v1.14 includes a connection upgrade API, which means it provides WebSocket
support out of the box. Let's see an example, this time using the Bandit webserver
and the websocket_adapter project for the WebSocket bits. Since we need different
routes, we will use the built-in Plug.Router for that:
Mix.install([:bandit, :websock_adapter])

defmodule EchoServer do
  def init(options) do
    {:ok, options}
  end

  def handle_in({"ping", [opcode: :text]}, state) do
    {:reply, :ok, {:text, "pong"}, state}
  end

  def terminate(:timeout, state) do
    {:ok, state}
  end
end

defmodule Router do
  use Plug.Router

  plug Plug.Logger
  plug :match
  plug :dispatch

  get "/" do
    send_resp(conn, 200, """
    Use the JavaScript console to interact using websockets

    sock  = new WebSocket("ws://localhost:4000/websocket")
    sock.addEventListener("message", console.log)
    sock.addEventListener("open", () => sock.send("ping"))
    """)
  end

  get "/websocket" do
    conn
    |> WebSockAdapter.upgrade(EchoServer, [], timeout: 60_000)
    |> halt()
  end

  match _ do
    send_resp(conn, 404, "not found")
  end
end

require Logger
webserver = {Bandit, plug: Router, scheme: :http, port: 4000}
{:ok, _} = Supervisor.start_link([webserver], strategy: :one_for_one)
Logger.info("Plug now running on localhost:4000")
Process.sleep(:infinity)
Save that snippet to a file and execute it as elixir websockets.exs.
Access http://localhost:4000/ and you should see messages in your browser
console.
As you can see, Plug abstracts the different webservers. When booting
up your application, the difference is between choosing Plug.Cowboy
or Bandit.
For now, we have directly started the server in a throw-away supervisor but,
for production deployments, you want to start them in application
supervision tree. See the Supervised handlers section next.

  
    
  
  Supervised handlers


On a production system, you likely want to start your Plug pipeline under your application's supervision tree. Start a new Elixir project with the --sup flag:
$ mix new my_app --sup

Add :plug_cowboy (or :bandit) as a dependency to your mix.exs:
def deps do
  [
    {:plug_cowboy, "~> 2.0"}
  ]
end
Now update lib/my_app/application.ex as follows:
defmodule MyApp.Application do
  # See https://hexdocs.pm/elixir/Application.html
  # for more information on OTP Applications
  @moduledoc false

  use Application

  def start(_type, _args) do
    # List all child processes to be supervised
    children = [
      {Plug.Cowboy, scheme: :http, plug: MyPlug, options: [port: 4001]}
    ]

    # See https://hexdocs.pm/elixir/Supervisor.html
    # for other strategies and supported options
    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end
end
Finally create lib/my_app/my_plug.ex with the MyPlug module.
Now run mix run --no-halt and it will start your application with a web server running at http://localhost:4001.

  
    
  
  Supported Versions


	Branch	Support
	v1.15	Bug fixes
	v1.14	Security patches only
	v1.13	Security patches only
	v1.12	Security patches only
	v1.11	Security patches only
	v1.10	Security patches only
	v1.9	Unsupported from 10/2023
	v1.8	Unsupported from 01/2023
	v1.7	Unsupported from 01/2022
	v1.6	Unsupported from 01/2022
	v1.5	Unsupported from 03/2021
	v1.4	Unsupported from 12/2018
	v1.3	Unsupported from 12/2018
	v1.2	Unsupported from 06/2018
	v1.1	Unsupported from 01/2018
	v1.0	Unsupported from 05/2017


  
    
  
  The Plug.Conn struct


In the hello world example, we defined our first plug. What is a plug after all?
A plug takes two shapes. A function plug receives a connection and a set of options as arguments and returns the connection:
def hello_world_plug(conn, _opts) do
  conn
  |> put_resp_content_type("text/plain")
  |> send_resp(200, "Hello world")
end
A module plug implements an init/1 function to initialize the options and a call/2 function which receives the connection and initialized options and returns the connection:
defmodule MyPlug do
  def init([]), do: false
  def call(conn, _opts), do: conn
end
As per the specification above, a connection is represented by the Plug.Conn struct:
%Plug.Conn{
  host: "www.example.com",
  path_info: ["bar", "baz"],
  ...
}
Data can be read directly from the connection and also pattern matched on. Manipulating the connection often happens with the use of the functions defined in the Plug.Conn module. In our example, both put_resp_content_type/2 and send_resp/3 are defined in Plug.Conn.
Remember that, as everything else in Elixir, a connection is immutable, so every manipulation returns a new copy of the connection:
conn = put_resp_content_type(conn, "text/plain")
conn = send_resp(conn, 200, "ok")
conn
Finally, keep in mind that a connection is a direct interface to the underlying web server. When you call send_resp/3 above, it will immediately send the given status and body back to the client. This makes features like streaming a breeze to work with.

  
    
  
  Plug.Router


To write a "router" plug that dispatches based on the path and method of incoming requests, Plug provides Plug.Router:
defmodule MyRouter do
  use Plug.Router

  plug :match
  plug :dispatch

  get "/hello" do
    send_resp(conn, 200, "world")
  end

  forward "/users", to: UsersRouter

  match _ do
    send_resp(conn, 404, "oops")
  end
end
The router is a plug. Not only that: it contains its own plug pipeline too. The example above says that when the router is invoked, it will invoke the :match plug, represented by a local (imported) match/2 function, and then call the :dispatch plug which will execute the matched code.
Plug ships with many plugs that you can add to the router plug pipeline, allowing you to plug something before a route matches or before a route is dispatched to. For example, if you want to add logging to the router, just do:
plug Plug.Logger
plug :match
plug :dispatch
Note Plug.Router compiles all of your routes into a single function and relies on the Erlang VM to optimize the underlying routes into a tree lookup, instead of a linear lookup that would instead match route-per-route. This means route lookups are extremely fast in Plug!
This also means that a catch all match block is recommended to be defined as in the example above, otherwise routing fails with a function clause error (as it would in any regular Elixir function).
Each route needs to return the connection as per the Plug specification. See the Plug.Router docs for more information.

  
    
  
  Testing plugs


Plug ships with a Plug.Test module that makes testing your plugs easy. Here is how we can test the router from above (or any other plug):
defmodule MyPlugTest do
  use ExUnit.Case, async: true
  use Plug.Test

  @opts MyRouter.init([])

  test "returns hello world" do
    # Create a test connection
    conn = conn(:get, "/hello")

    # Invoke the plug
    conn = MyRouter.call(conn, @opts)

    # Assert the response and status
    assert conn.state == :sent
    assert conn.status == 200
    assert conn.resp_body == "world"
  end
end

  
    
  
  Available plugs


This project aims to ship with different plugs that can be re-used across applications:
	Plug.BasicAuth - provides Basic HTTP authentication;
	Plug.CSRFProtection - adds Cross-Site Request Forgery protection to your application. Typically required if you are using Plug.Session;
	Plug.Head - converts HEAD requests to GET requests;
	Plug.Logger - logs requests;
	Plug.MethodOverride - overrides a request method with one specified in the request parameters;
	Plug.Parsers - responsible for parsing the request body given its content-type;
	Plug.RequestId - sets up a request ID to be used in logs;
	Plug.RewriteOn - rewrite the request's host/port/protocol from x-forwarded-* headers;
	Plug.Session - handles session management and storage;
	Plug.SSL - enforces requests through SSL;
	Plug.Static - serves static files;
	Plug.Telemetry - instruments the plug pipeline with :telemetry events;

You can go into more details about each of them in our docs.

  
    
  
  Helper modules


Modules that can be used after you use Plug.Router or Plug.Builder to help development:
	Plug.Debugger - shows a helpful debugging page every time there is a failure in a request;
	Plug.ErrorHandler - allows developers to customize error pages in case of crashes instead of sending a blank one;


  
    
  
  Contributing


We welcome everyone to contribute to Plug and help us tackle existing issues!
Use the issue tracker for bug reports or feature requests. Open a pull request when you are ready to contribute. When submitting a pull request you should not update the CHANGELOG.md.
If you are planning to contribute documentation, please check our best practices for writing documentation.
Finally, remember all interactions in our official spaces follow our Code of Conduct.

  
    
  
  License


Plug source code is released under Apache License 2.0.
Check LICENSE file for more information.


  

    
HTTPS
    

Plug can serve HTTP over TLS ('HTTPS') through an appropriately configured Adapter. While the exact syntax for defining an HTTPS listener is adapter-specific, Plug does define a common set of TLS configuration options that most adapters support, formally documented as Plug.SSL.configure/1.
This guide describes how to use these parameters to set up an HTTPS server with Plug, and documents some best-practices and potential pitfalls.
Editor's note: The secure transport protocol used by HTTPS is nowadays referred to as TLS. However, the application in the Erlang/OTP standard library that implements it is called :ssl, for historical reasons. In this document we will refer to the protocol as 'TLS' and to the Erlang/OTP implementation as :ssl, and its configuration parameters as :ssl options.


  
    
  
  Prerequisites


The prerequisites for running an HTTPS server with Plug include:
	The Erlang/OTP runtime, with OpenSSL bindings; run :crypto.info_lib() in an IEx session to verify
	A Plug Adapter that supports HTTPS, e.g. Plug.Cowboy
	A valid certificate and associated private key


  
    
  
  Self-signed Certificate


For testing purposes it may be sufficient to use a self-signed certificate. Such certificates generally result in warnings in browsers and failed connections from other tools, but these can be overridden to enable HTTPS testing. This is especially useful for local testing of HTTP 2, which is only specified over TLS.
Warning: use self-signed certificates only for local testing, and do not mark such test certificates as globally trusted in browsers or operating system!

The Phoenix project includes a Mix task mix phx.gen.cert that generates the necessary files and places them in the application's 'priv' directory. The X509 package can be used as a dev-only dependency to add a similar mix x509.gen.selfsigned task to non-Phoenix projects.
Alternatively, the OpenSSL CLI or other utilities can be used to generate a self-signed certificate. Instructions are widely available online.

  
    
  
  CA Issued Certificate


For staging and production it is necessary to obtain a CA-signed certificate from a trusted Certificate Authority, such as Let's Encrypt. Certificates issued by a CA usually come with an additional file containing one or more certificates that make up the 'CA chain'.
For use with Plug the certificates and key should be stored in PEM format, containing Base64-encoded data between 'BEGIN' and 'END' markers. Some useful OpenSSL commands for converting certificates/keys from other formats can be found at the end of this document.

  
    
  
  Getting Started


A minimal HTTPS listener, using Plug.Cowboy, might be defined as follows:
Plug.Cowboy.https MyApp.MyPlug, [],
  port: 8443,
  cipher_suite: :strong,
  certfile: "/etc/letsencrypt/live/example.net/cert.pem",
  keyfile: "/etc/letsencrypt/live/example.net/privkey.pem",
  cacertfile: "/etc/letsencrypt/live/example.net/chain.pem"
The cacertfile option is not needed when using a self-signed certificate, or when the file pointed to by certfile contains both the server certificate and all necessary CA chain certificates:
  #...
  certfile: "/etc/letsencrypt/live/example.net/fullchain.pem",
  keyfile: "/etc/letsencrypt/live/example.net/privkey.pem"
It is possible to bundle the certificate files with the application, possibly for packaging into a release. In this case the files must be stored under the application's 'priv' directory. The otp_app option must be set to the name of the OTP application that contains the files, in order to correctly resolve the relative paths:
Plug.Cowboy.https MyApp.MyPlug, [],
  port: 8443,
  cipher_suite: :strong,
  certfile: "priv/cert/selfsigned.pem",
  keyfile: "priv/cert/selfsigned_key.pem",
  otp_app: :my_app
Remember to exclude the files from version control, unless the certificate and key are shared by all developers for testing purposes only. For example, add this line to the '.gitignore' file: priv/**/*.pem.

  
    
  
  TLS Protocol Options


In addition to a certificate, an HTTPS server needs a secure TLS protocol configuration. Plug.SSL always sets the following options:
	Set secure_renegotiate: true, to avoid certain types of man-in-the-middle attacks
	Set reuse_sessions: true, for improved handshake performance of recurring connections

Additional options can be set by selecting a predefined profile or by setting :ssl options individually.

  
    
  
  Predefined Options


To simplify configuration of TLS defaults Plug provides two preconfigured options: cipher_suite: :strong and cipher_suite: :compatible.
The :strong profile enables AES-GCM ciphers with ECDHE or DHE key exchange, and TLS version 1.2 only. It is intended for typical installations with support for browsers and other modern clients.
The :compatible profile additionally enables AES-CBC ciphers, as well as TLS versions 1.1 and 1.0. Use this configuration to allow connections from older clients, such as older PC or mobile operating systems. Note that RSA key exchange is not enabled by this configuration, due to known weaknesses, so to support clients that do not support ECDHE or DHE it is necessary specify the ciphers explicitly (see below).
In addition, both profiles:
	Configure the server to choose a cipher based on its own preferences rather than the client's (honor_cipher_order set to true); when specifying a custom cipher list, ensure the ciphers are listed in descending order of preference
	Select the 'Prime' (SECP) curves for use in Elliptic Curve Cryptography (ECC)

All these parameters, including the global defaults mentioned above, can be overridden by specifying custom :ssl configuration options.
It is worth noting that the cipher lists and TLS protocol versions selected by the profiles are whitelists. If a new Erlang/OTP release introduces new TLS protocol versions or ciphers that are not included in the profile definition, they would have to be enabled explicitly by overriding the :ciphers and/or :versions options, until such time as they are added to the Plug.SSL profiles.
The ciphers chosen and related configuration are based on OWASP recommendations, with some modifications as described in the Plug.SSL.configure/1 documentation.

  
    
  
  Manual Configuration


Please refer to the Erlang/OTP :ssl documentation for details on the supported configuration options.
An example configuration with custom :ssl options might look like this:
Plug.Cowboy.https MyApp.MyPlug, [],
  port: 8443,
  certfile: "/etc/letsencrypt/live/example.net/cert.pem",
  keyfile: "/etc/letsencrypt/live/example.net/privkey.pem",
  cacertfile: "/etc/letsencrypt/live/example.net/chain.pem",
  versions: [:"tlsv1.2", :"tlsv1.1"],
  ciphers: [
    'ECDHE-RSA-AES256-GCM-SHA384',
    'ECDHE-RSA-AES128-GCM-SHA256',
    'DHE-RSA-AES256-GCM-SHA384',
    'DHE-RSA-AES128-GCM-SHA256'
  ],
  honor_cipher_order: true,
  sni_fun: &MyPlug.ssl_opts_for_hostname/1

  
    
  
  HTTP Strict Transport Security (HSTS)


Once a server is configured to support HTTPS it is often a good idea to redirect HTTP requests to HTTPS. To do this, include Plug.SSL in the Plug pipeline.
To prevent downgrade attacks, in which an attacker intercepts a plain HTTP request to the server before the redirect to HTTPS takes place, Plug.SSL by default sets the 'Strict-Transport-Security' (HSTS) header. This informs the browser that the current site must only ever be accessed over HTTPS, even if the user typed or clicked a plain HTTP URL. This only works if the site is reachable on port 443 (see Listening on Port 443, below).
Warning: it is very difficult, if not impossible, to revert the effect of HSTS before the entry stored in the browser expires! Consider using a short :expires value initially, and increasing it to a large value (e.g. 31536000 seconds for 1 year) after testing.

The Strict-Transport-Security header can be disabled altogether by setting hsts: false in the Plug.SSL options.

  
    
  
  Encrypted Keys


To protect the private key on disk it is best stored in encrypted PEM format, protected by a password. When configuring a Plug server with an encrypted private key, specify the password using the :password option:
Plug.Cowboy.https MyApp.MyPlug, [],
  port: 8443,
  certfile: "/etc/letsencrypt/live/example.net/cert.pem",
  keyfile: "/etc/letsencrypt/live/example.net/privkey_aes.pem",
  cacertfile: "/etc/letsencrypt/live/example.net/chain.pem",
  password: "SECRET"
To encrypt an existing PEM-encoded RSA key use the OpenSSL CLI: openssl rsa -in privkey.pem -out privkey_aes.pem -aes128. Use ec instead of rsa when using an ECDSA certificate. Don't forget to securely erase the unencrypted copy afterwards! Best practice would be to encrypt the file immediately during initial key generation: please refer to the instructions provided by the CA.
Note: at the time of writing, Erlang/OTP does not support keys encrypted with AES-256. The OpenSSL command in the previous paragraph can also be used to convert an AES-256 encrypted key to AES-128.


  
    
  
  Passing DER Binaries


Sometimes it is preferable to not store the private key on disk at all. Instead, the private key might be passed to the application using an environment variable or retrieved from a key store such as Vault.
In such cases it is possible to pass the private key directly, using the :key parameter. For example, assuming an RSA private key is available in the PRIVKEY environment variable in Base64 encoded DER format, the key may be set as follows:
der = System.get_env("PRIVKEY") |> Base.decode64!
Plug.Cowboy.https MyApp.MyPlug, [],
  port: 8443,
  key: {:RSAPrivateKey, der},
  #....
Note that reading environment variables in Mix config files only works when starting the application using Mix, e.g. in a development environment. In production, a different approach is needed for runtime configuration, but this is out of scope for the current document.
The certificate and CA chain can also be specified using DER binaries, using the :cert and :cacerts options, but this is best avoided. The use of PEM files has been tested much more thoroughly with the Erlang/OTP :ssl application, and there have been a number of issues with DER binary certificates in the past.

  
    
  
  Custom Diffie-Hellman Parameters


It is recommended to generate a custom set of Diffie-Hellman parameters, to be used for the DHE key exchange. Use the following OpenSSL CLI command to create a dhparam.pem file:
openssl dhparam -out dhparams.pem 4096
On a slow machine (e.g. a cheap VPS) this may take several hours. You may want to run the command on a strong machine and copy the file over to the target server: the file does not need to be kept secret. It is best practice to rotate the file periodically.
Pass the (relative or absolute) path using the :dhfile option:
Plug.Cowboy.https MyApp.MyPlug, [],
  port: 8443,
  cipher_suite: :strong,
  certfile: "priv/cert/selfsigned.pem",
  keyfile: "priv/cert/selfsigned_key.pem",
  dhfile: "priv/cert/dhparams.pem",
  otp_app: :my_app
If no custom parameters are specified, Erlang's :ssl uses its built-in defaults. Since OTP 19 this has been the 2048-bit 'group 14' from RFC3526.

  
    
  
  Renewing Certificates


Whenever a certificate is about to expire, when the contents of the certificate have been updated, or when the certificate is 're-keyed', the HTTPS server needs to be updated with the new certificate and/or key.
When using the :certfile and :keyfile parameters to reference PEM files on disk, replacing the certificate and key is as simple as overwriting the files. Erlang's :ssl application periodically reloads any referenced files, with changes taking effect in subsequent handshakes. It may be best to use symbolic links that point to versioned copies of the files, to allow for quick rollback in case of problems.
Note that there is a potential race condition when both the certificate and the key need to be replaced at the same time: if the :ssl application reloads one file before the other file is updated, the partial update can leave the HTTPS server with a mismatched private key. This can be avoiding by placing the private key in the same PEM file as the certificate, and omitting the :keyfile option. This configuration allows atomic updates, and it works because :ssl looks for a private key entry in the :certfile PEM file if no :key or :keyfile option is specified.
While it is possible to update the DER binaries passed in the :cert or :key options (as well as any other TLS protocol parameters) at runtime, this requires knowledge of the internals of the Plug adapter being used, and is therefore beyond the scope of this document.

  
    
  
  Listening on Port 443


By default clients expect HTTPS servers to listen on port 443. It is possible to specify a different port in HTTPS URLs, but for public servers it is often preferable to stick to the default. In particular, HSTS requires that the site be reachable on port 443 using HTTPS.
This presents a problem, however: only privileged processes can bind to TCP port numbers under 1024, and it is bad idea to run the application as 'root'.
Leaving aside solutions that rely on external network elements, such as load balancers, there are several solutions on typical Linux servers:
	Deploy a reverse proxy or load balancer process, such as Nginx or HAProxy (see Offloading TLS, below); the proxy listens on port 443 and passes the traffic to the Elixir application running on an unprivileged port
	Create an IPTables rule to forward packets arriving on port 443 to the port on which the Elixir application is running
	Give the Erlang/OTP runtime (that is, the BEAM VM executable) permission to bind to privileged ports using 'setcap', e.g. sudo setcap 'cap_net_bind_service=+ep' /usr/lib/erlang/erts-10.1/bin/beam.smp; update the path as necessary, and remember to run the command again after Erlang upgrades
	Use a tool such as 'authbind' to give an unprivileged user/group permission to bind to specific ports

This is not intended to be an exhaustive list, as this topic is actually a bit beyond the scope of the current document. The issue is a generic one, not specific to Erlang/Elixir, and further explanations can be found online.

  
    
  
  Offloading TLS


So far this document has focused on configuring Plug to handle TLS within the application. Some people instead prefer to terminate TLS in a proxy or load balancer deployed in front of the Plug application.

  
    
  
  Pros and Cons


Offloading might be done to achieve higher throughput, or to stick to the more widely used OpenSSL implementation of the TLS protocol. The Erlang/OTP implementation depends on OpenSSL for the underlying cryptography, but it implements its own message framing and protocol state machine. While it is not clear that one implementation is inherently more secure than the other, just patching OpenSSL along with everybody else in case of vulnerabilities might give peace of mind, compared to than having to research the implications on the Erlang/OTP implementation.
On the other hand, the proxy solution might not support end-to-end HTTP 2, limiting the benefits of the new protocol. It can also introduce operational complexities and new resource constraints, especially for long-lived connections such as WebSockets.

  
    
  
  Plug Configuration Impact


When using TLS offloading it may be necessary to make some configuration changes to the application.
Plug.SSL takes on another important role when using TLS offloading: it can update the :scheme and :port fields in the Plug.Conn struct based on an HTTP header (e.g. 'X-Forwarded-Proto'), to reflect the actual protocol used by the client (HTTP or HTTPS). It is very important that the :scheme field properly reflects the use of HTTPS, even if the connection between the proxy and the application uses plain HTTP, because cookies set by Plug.Session and Plug.Conn.put_resp_cookie/4 by default set the 'secure' cookie flag only if :scheme is set to :https! When relying on this default behaviour it is essential that Plug.SSL is included in the Plug pipeline, that its :rewrite_on option is set correctly, and that the proxy sets the appropriate header.
The :remote_ip field in the Plug.Conn struct by default contains the network peer IP address. Terminating TLS in a separate process or network element typically masks the actual client IP address from the Elixir application. If proxying is done at the HTTP layer, the original client IP address is often inserted into an HTTP header, e.g. 'X-Forwarded-For'. There are Plugs available to extract the client IP from such a header, such as Plug.RewriteOn.
Warning: ensure that clients cannot spoof their IP address by including this header in their original request, by filtering such headers in the proxy!

For solutions that operate below the HTTP layer, e.g. using HAProxy, the client IP address can sometimes be passed through the 'PROXY protocol'. Extracting this information must be handled by the Plug adapter. Please refer to the Plug adapter documentation for further information.

  
    
  
  Converting Certificates and Keys


When certificate and/or key files are not in provided in PEM format they can usually be converted using the OpenSSL CLI. This section describes some common formats and the associated OpenSSL commands to convert to PEM.

  
    
  
  From DER to PEM


DER-encoded files contain binary data. Common file extensions are .crt for certificates and .key for keys.
To convert a single DER-encoded certificate to PEM format: openssl x509 -in server.crt -inform der -out cert.pem
To convert an RSA private key from DER to PEM format: openssl rsa -in privkey.der -inform der -out privkey.pem. If the private key is a Elliptic Curve key, for use with an ECDSA certificate, replace rsa with ec. You may want to add the -aes128 argument to produce an encrypted, password protected PEM file.

  
    
  
  From PKCS#12 to PEM


The PKCS#12 format is a container format containing one or more certificates and/or encrypted keys. Such files typically have a .p12 extension.
To extract all certificates from a PKCS#12 file to a PEM file: openssl pkcs12 -in server.p12 -nokeys -out fullchain.pem. The resulting file contains all certificates from the input file, typically the server certificate and any CA certificates that make up the CA chain. You can split the file into separate cert.pem and chain.pem files using a text editor, or you can just pass certfile: fullchain.pem to the HTTPS adapter.
To extract a private key from a PKCS#12 file to a PEM file: openssl pkcs12 -in server.p12 -nocerts -nodes -out privkey.pem. You may want to replace the -nodes argument with -aes128 to produce an encrypted, password protected PEM file.

The plug specification.

  
    
  
  Types of plugs


There are two kind of plugs: function plugs and module plugs.

  
    
  
  Function plugs


A function plug is by definition any function that receives a connection
and a set of options and returns a connection. Function plugs must have
the following type signature:
(Plug.Conn.t, Plug.opts) :: Plug.Conn.t

  
    
  
  Module plugs


A module plug is an extension of the function plug. It is a module that must
export:
	a call/2 function with the signature defined above
	an init/1 function which takes a set of options and initializes it.

The result returned by init/1 is passed as second argument to call/2. Note
that init/1 may be called during compilation and as such it must not return
pids, ports or values that are specific to the runtime.
The API expected by a module plug is defined as a behaviour by the
Plug module (this module).

  
    
  
  Examples


Here's an example of a function plug:
def json_header_plug(conn, _opts) do
  Plug.Conn.put_resp_content_type(conn, "application/json")
end
Here's an example of a module plug:
defmodule JSONHeaderPlug do
  @behaviour Plug

  import Plug.Conn

  def init(opts) do
    opts
  end

  def call(conn, _opts) do
    put_resp_content_type(conn, "application/json")
  end
end

  
    
  
  The Plug pipeline


The Plug.Builder module provides conveniences for building plug pipelines.

Conveniences for building plugs.
You can use this module to build a plug pipeline:
defmodule MyApp do
  use Plug.Builder

  plug Plug.Logger
  plug :hello, upper: true

  # A function from another module can be plugged too, provided it's
  # imported into the current module first.
  import AnotherModule, only: [interesting_plug: 2]
  plug :interesting_plug

  def hello(conn, opts) do
    body = if opts[:upper], do: "WORLD", else: "world"
    send_resp(conn, 200, body)
  end
end
The plug/2 macro forms a pipeline by defining multiple plugs. Each plug
in the pipeline is executed from top to bottom. In the example above, the
Plug.Logger module plug is called before the :hello function plug, so
the function plug will be called on the module plug's resulting connection.
Plug.Builder imports the Plug.Conn module so functions like send_resp/3
are available.

  
    
  
  Options


When used, the following options are accepted by Plug.Builder:
	:init_mode - the environment to initialize the plug's options, one of
:compile or :runtime. The default value is :compile.

	:log_on_halt - accepts the level to log whenever the request is halted

	:copy_opts_to_assign - an atom representing an assign. When supplied,
it will copy the options given to the Plug initialization to the given
connection assign



  
    
  
  Plug behaviour


Plug.Builder defines the init/1 and call/2 functions by implementing
the Plug behaviour.
By implementing the Plug API, Plug.Builder guarantees this module is a plug
and can be handed to a web server or used as part of another pipeline.

  
    
  
  Conditional plugs


Sometimes you may want to conditionally invoke a Plug in a pipeline. For example,
you may want to invoke Plug.Parsers only under certain routes. This can be done
by wrapping the module plug in a function plug. Instead of:
plug Plug.Parsers, parsers: [:urlencoded, :multipart], pass: ["text/*"]
You can write:
plug :conditional_parser

defp conditional_parser(%Plug.Conn{path_info: ["noparser" | _]} = conn, _opts) do
  conn
end

@parser Plug.Parsers.init(parsers: [:urlencoded, :multipart], pass: ["text/*"])
defp conditional_parser(conn, _opts) do
  Plug.Parsers.call(conn, @parser)
end
The above will invoke Plug.Parsers on all routes, except the ones under /noparser

  
    
  
  Overriding the default Plug API functions


Both the init/1 and call/2 functions defined by Plug.Builder can be
manually overridden. For example, the init/1 function provided by
Plug.Builder returns the options that it receives as an argument, but its
behaviour can be customized:
defmodule PlugWithCustomOptions do
  use Plug.Builder
  plug Plug.Logger

  def init(opts) do
    opts
  end
end
The call/2 function that Plug.Builder provides is used internally to
execute all the plugs listed using the plug macro, so overriding the
call/2 function generally implies using super in order to still call the
plug chain:
defmodule PlugWithCustomCall do
  use Plug.Builder
  plug Plug.Logger
  plug Plug.Head

  def call(conn, opts) do
    conn
    |> super(opts) # calls Plug.Logger and Plug.Head
    |> assign(:called_all_plugs, true)
  end
end

  
    
  
  Halting a plug pipeline


Plug.Conn.halt/1 halts a plug pipeline. Plug.Builder prevents plugs
downstream from being invoked and returns the current connection.
In the following example, the Plug.Logger plug never gets
called:
defmodule PlugUsingHalt do
  use Plug.Builder

  plug :stopper
  plug Plug.Logger

  def stopper(conn, _opts) do
    halt(conn)
  end
end

The Plug connection.
This module defines a struct and the main functions for working with
requests and responses in an HTTP connection.
Note request headers are normalized to lowercase and response
headers are expected to have lowercase keys.

  
    
  
  Request fields


These fields contain request information:
	host - the requested host as a binary, example: "www.example.com"
	method - the request method as a binary, example: "GET"
	path_info - the path split into segments, example: ["hello", "world"]
	script_name - the initial portion of the URL's path that corresponds to
the application routing, as segments, example: ["sub","app"]
	request_path - the requested path, example: /trailing/and//double//slashes/
	port - the requested port as an integer, example: 80
	remote_ip - the IP of the client, example: {151, 236, 219, 228}. This field
is meant to be overwritten by plugs that understand e.g. the X-Forwarded-For
header or HAProxy's PROXY protocol. It defaults to peer's IP
	req_headers - the request headers as a list, example: [{"content-type", "text/plain"}].
Note all headers will be downcased
	scheme - the request scheme as an atom, example: :http
	query_string - the request query string as a binary, example: "foo=bar"


  
    
  
  Fetchable fields


Fetchable fields do not populate with request information until the corresponding
prefixed 'fetch_' function retrieves them, e.g., the fetch_cookies/2 function
retrieves the cookies field.
If you access these fields before fetching them, they will be returned as
Plug.Conn.Unfetched structs.
	cookies- the request cookies with the response cookies
	body_params - the request body params, populated through a Plug.Parsers parser.
	query_params - the request query params, populated through fetch_query_params/2
	path_params - the request path params, populated by routers such as Plug.Router
	params - the request params, the result of merging the :path_params on top of
 :body_params on top of :query_params
	req_cookies - the request cookies (without the response ones)


  
    
  
  Session vs Assigns


HTTP is stateless.
This means that a server begins each request cycle with no knowledge about
the client except the request itself. Its response may include one or more
"Set-Cookie" headers, asking the client to send that value back in a
"Cookie" header on subsequent requests.
This is the basis for stateful interactions with a client, so that the server
can remember the client's name, the contents of their shopping cart, and so on.
In Plug, a "session" is a place to store data that persists from one request
to the next. Typically, this data is stored in a cookie using Plug.Session.COOKIE.
A minimal approach would be to store only a user's id in the session, then
use that during the request cycle to look up other information (in a database
or elsewhere).
More can be stored in a session cookie, but be careful: this makes requests
and responses heavier, and clients may reject cookies beyond a certain size.
Also, session cookie are not shared between a user's different browsers or devices.
If the session is stored elsewhere, such as a database, the browser only has to
store the session key and therefore more data can be stored in the session.
A typical use case would be for an authentication plug to look up a user by id
and keep the user information stored in assigns. Other plugs will then also
have access to it via assigns. This is an important point because the assign
data disappears on the next request.
To summarize: assigns is for storing data to be accessed during the current
request, and the session is for storing data to be accessed in subsequent
requests.

  
    
  
  Response fields


These fields contain response information:
	resp_body - the response body is an empty string by default. It is set
to nil after the response is sent, except for test connections. The response
charset defaults to "utf-8".
	resp_cookies - the response cookies with their name and options
	resp_headers - the response headers as a list of tuples, cache-control
is set to "max-age=0, private, must-revalidate" by default.
Note: Use all lowercase for response headers.
	status - the response status


  
    
  
  Connection fields


	assigns - shared user data as a map
	owner - the Elixir process that owns the connection
	halted - the boolean status on whether the pipeline was halted
	secret_key_base - a secret key used to verify and encrypt cookies.
These features require manual field setup. Data must be kept in the
connection and never used directly. Always use Plug.Crypto.KeyGenerator.generate/3
to derive keys from it.
	state - the connection state

The connection state is used to track the connection lifecycle. It starts as
:unset but is changed to :set (via resp/3) or :set_chunked
(used only for before_send callbacks by send_chunked/2) or :file
(when invoked via send_file/3). Its final result is :sent, :file, :chunked
or :upgraded depending on the response model.

  
    
  
  Private fields


These fields are reserved for libraries/framework usage.
	adapter - holds the adapter information in a tuple
	private - shared library data as a map


  
    
  
  Custom status codes


Plug allows status codes to be overridden or added and allow new codes not directly
specified by Plug or its adapters. The :plug application's Mix config can add or
override a status code.
For example, the config below overrides the default 404 reason phrase ("Not Found")
and adds a new 998 status code:
config :plug, :statuses, %{
  404 => "Actually This Was Found",
  998 => "Not An RFC Status Code"
}
Dependency-specific config changes are not automatically recompiled. Recompile Plug
for the changes to take place. The command below recompiles Plug:
mix deps.clean --build plug
A corresponding atom is inflected from each status code reason phrase. In many functions,
these atoms can stand in for the status code. For example, with the above configuration,
the following will work:
put_status(conn, :not_found)                     # 404
put_status(conn, :actually_this_was_found)       # 404
put_status(conn, :not_an_rfc_status_code)        # 998
The :not_found atom can still be used to set the 404 status even though the 404 status code
reason phrase was overwritten. The new atom :actually_this_was_found, inflected from the
reason phrase "Actually This Was Found", can also be used to set the 404 status code.

  
    
  
  Protocol Upgrades


Plug.Conn.upgrade_adapter/3 provides basic support for protocol upgrades and facilitates
connection upgrades to protocols such as WebSockets. As the name suggests, this functionality
is adapter-dependent. Protocol upgrade functionality requires explicit coordination between
a Plug application and the underlying adapter.
Plug upgrade-related functionality only provides the possibility for the Plug application
to request protocol upgrades from the underlying adapter. See upgrade_adapter/3 documentation.

Conveniences for generating HTML.

A DSL to define a routing algorithm that works with Plug.
It provides a set of macros to generate routes. For example:
defmodule AppRouter do
  use Plug.Router

  plug :match
  plug :dispatch

  get "/hello" do
    send_resp(conn, 200, "world")
  end

  match _ do
    send_resp(conn, 404, "oops")
  end
end
Each route receives a conn variable containing a Plug.Conn
struct and it needs to return a connection, as per the Plug spec.
A catch-all match is recommended to be defined as in the example
above, otherwise routing fails with a function clause error.
The router is itself a plug, which means it can be invoked as:
AppRouter.call(conn, AppRouter.init([]))
Each Plug.Router has a plug pipeline, defined by Plug.Builder,
and by default it requires two plugs: :match and :dispatch.
:match is responsible for finding a matching route which is
then forwarded to :dispatch. This means users can easily hook
into the router mechanism and add behaviour before match, before
dispatch, or after both. See the Plug.Builder module for more
information.

  
    
  
  Routes


get "/hello" do
  send_resp(conn, 200, "world")
end
In the example above, a request will only match if it is a GET
request and the route is "/hello". The supported HTTP methods are
get, post, put, patch, delete and options.
A route can also specify parameters which will then be available
in the function body:
get "/hello/:name" do
  send_resp(conn, 200, "hello #{name}")
end
This means the name can also be used in guards:
get "/hello/:name" when name in ~w(foo bar) do
  send_resp(conn, 200, "hello #{name}")
end
The :name parameter will also be available in the function body as
conn.params["name"] and conn.path_params["name"].
The identifier always starts with : and must be followed by letters,
numbers, and underscores, like any Elixir variable. It is possible for
identifiers to be either prefixed or suffixed by other words. For example,
you can include a suffix such as a dot delimited file extension:
get "/hello/:name.json" do
  send_resp(conn, 200, "hello #{name}")
end
The above will match /hello/foo.json but not /hello/foo.
Other delimiters such as -, @ may be used to denote suffixes.
Routes allow for globbing which will match the remaining parts
of a route. A glob match is done with the * character followed
by the variable name. Typically you prefix the variable name with
underscore to discard it:
get "/hello/*_rest" do
  send_resp(conn, 200, "matches all routes starting with /hello")
end
But you can also assign the glob to any variable. The contents will
always be a list:
get "/hello/*glob" do
  send_resp(conn, 200, "route after /hello: #{inspect glob}")
end
Opposite to :identifiers, globs do not allow prefix nor suffix
matches.
Finally, a general match function is also supported:
match "/hello" do
  send_resp(conn, 200, "world")
end
A match will match any route regardless of the HTTP method.
Check match/3 for more information on how route compilation
works and a list of supported options.

  
    
  
  Parameter Parsing


Handling request data can be done through the
Plug.Parsers plug. It
provides support for parsing URL-encoded, form-data, and JSON data as well as
providing a behaviour that others parsers can adopt.
Here is an example of Plug.Parsers can be used in a Plug.Router router to
parse the JSON-encoded body of a POST request:
defmodule AppRouter do
  use Plug.Router

  plug :match

  plug Plug.Parsers,
       parsers: [:json],
       pass:  ["application/json"],
       json_decoder: Jason

  plug :dispatch

  post "/hello" do
    IO.inspect conn.body_params # Prints JSON POST body
    send_resp(conn, 200, "Success!")
  end
end
It is important that Plug.Parsers is placed before the :dispatch plug in
the pipeline, otherwise the matched clause route will not receive the parsed
body in its Plug.Conn argument when dispatched.
Plug.Parsers can also be plugged between :match and :dispatch (like in
the example above): this means that Plug.Parsers will run only if there is a
matching route. This can be useful to perform actions such as authentication
before parsing the body, which should only be parsed if a route matches
afterwards.

  
    
  
  Error handling


In case something goes wrong in a request, the router by default
will crash, without returning any response to the client. This
behaviour can be configured in two ways, by using two different
modules:
	Plug.ErrorHandler - allows the developer to customize exactly
which page is sent to the client via the handle_errors/2 function;

	Plug.Debugger - automatically shows debugging and request information
about the failure. This module is recommended to be used only in a
development environment.


Here is an example of how both modules could be used in an application:
defmodule AppRouter do
  use Plug.Router

  if Mix.env == :dev do
    use Plug.Debugger
  end

  use Plug.ErrorHandler

  plug :match
  plug :dispatch

  get "/hello" do
    send_resp(conn, 200, "world")
  end

  defp handle_errors(conn, %{kind: _kind, reason: _reason, stack: _stack}) do
    send_resp(conn, conn.status, "Something went wrong")
  end
end

  
    
  
  Passing data between routes and plugs


It is also possible to assign data to the Plug.Conn that will
be available to any plug invoked after the :match plug.
This is very useful if you want a matched route to customize how
later plugs will behave.
You can use :assigns (which contains user data) or :private
(which contains library/framework data) for this. For example:
get "/hello", assigns: %{an_option: :a_value} do
  send_resp(conn, 200, "world")
end
In the example above, conn.assigns[:an_option] will be available
to all plugs invoked after :match. Such plugs can read from
conn.assigns (or conn.private) to configure their behaviour
based on the matched route.

  
    
  
  use options


All of the options given to use Plug.Router are forwarded to
Plug.Builder. See the Plug.Builder module for more information.

  
    
  
  Telemetry


The router emits the following telemetry events:
	[:plug, :router_dispatch, :start] - dispatched before dispatching to a matched route
	Measurement: %{system_time: System.system_time}
	Metadata: %{telemetry_span_context: term(), conn: Plug.Conn.t, route: binary, router: module}


	[:plug, :router_dispatch, :exception] - dispatched after exceptions on dispatching a route
	Measurement: %{duration: native_time}
	Metadata: %{telemetry_span_context: term(), conn: Plug.Conn.t, route: binary, router: module, kind: :throw | :error | :exit, reason: term(), stacktrace: list()}



	[:plug, :router_dispatch, :stop] - dispatched after successfully dispatching a matched route
	Measurement: %{duration: native_time}
	Metadata: %{telemetry_span_context: term(), conn: Plug.Conn.t, route: binary, router: module}

Conveniences for testing plugs.
This module can be used in your test cases, like this:
use ExUnit.Case, async: true
use Plug.Test
Using this module will:
	import all the functions from this module
	import all the functions from the Plug.Conn module

By default, Plug tests checks for invalid header keys, e.g. header keys which
include uppercase letters, and raises a Plug.Conn.InvalidHeaderError when
it finds one. To disable it, set :validate_header_keys_during_test to
false on the app config.
config :plug, :validate_header_keys_during_test, false

A server (a GenServer specifically) that manages uploaded files.
Uploaded files are stored in a temporary directory
and removed from that directory after the process that
requested the file dies.
During the request, files are represented with
a Plug.Upload struct that contains three fields:
	:path - the path to the uploaded file on the filesystem
	:content_type - the content type of the uploaded file
	:filename - the filename of the uploaded file given in the request

Note: as mentioned in the documentation for Plug.Parsers, the :plug
application has to be started in order to upload files and use the
Plug.Upload module.

  
    
  
  Security


The :content_type and :filename fields in the Plug.Upload struct are
client-controlled. These values should be validated, via file content
inspection or similar, before being trusted.

Functionality for providing Basic HTTP authentication.
It is recommended to only use this module in production
if SSL is enabled and enforced. See Plug.SSL for more
information.

  
    
  
  Compile-time usage


If you have a single username and password, you can use
the basic_auth/2 plug:
import Plug.BasicAuth
plug :basic_auth, username: "hello", password: "secret"
Or if you would rather put those in a config file:
# lib/your_app.ex
import Plug.BasicAuth
plug :basic_auth, Application.compile_env(:my_app, :basic_auth)

# config/config.exs
config :my_app, :basic_auth, username: "hello", password: "secret"
Once the user first accesses the page, the request will be denied
with reason 401 and the request is halted. The browser will then
prompt the user for username and password. If they match, then the
request succeeds.
Both approaches shown above rely on static configuration. Let's see
alternatives.

  
    
  
  Runtime-time usage


As any other Plug, we can use the basic_auth at runtime by simply
wrapping it in a function:
plug :auth

defp auth(conn, _opts) do
  username = System.fetch_env!("AUTH_USERNAME")
  password = System.fetch_env!("AUTH_PASSWORD")
  Plug.BasicAuth.basic_auth(conn, username: username, password: password)
end
This approach is useful when both username and password are specified
upfront and available at runtime. However, you may also want to compute
a different password for each different user. In those cases, we can use
the low-level API.

  
    
  
  Low-level usage


If you want to provide your own authentication logic on top of Basic HTTP
auth, you can use the low-level functions. As an example, we define :auth
plug that extracts username and password from the request headers, compares
them against the database, and either assigns a :current_user on success
or responds with an error on failure.
plug :auth

defp auth(conn, _opts) do
  with {user, pass} <- Plug.BasicAuth.parse_basic_auth(conn),
       %User{} = user <- MyApp.Accounts.find_by_username_and_password(user, pass) do
    assign(conn, :current_user, user)
  else
    _ -> conn |> Plug.BasicAuth.request_basic_auth() |> halt()
  end
end
Keep in mind that:
	The supplied user and pass may be empty strings;

	If you are comparing the username and password with existing strings,
do not use ==/2 or pattern matching. Use Plug.Crypto.secure_compare/2
instead.



      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Plug.CSRFProtection - Plug v1.16.1
    
    

    


  
  

    
Plug.CSRFProtection 
    



      
Plug to protect from cross-site request forgery.
For this plug to work, it expects a session to have been
previously fetched. It will then compare the token stored
in the session with the one sent by the request to determine
the validity of the request. For an invalid request the action
taken is based on the :with option.
The token may be sent by the request either via the params
with key "_csrf_token" or a header with name "x-csrf-token".
GET requests are not protected, as they should not have any
side-effect or change your application state. JavaScript
requests are an exception: by using a script tag, external
websites can embed server-side generated JavaScript, which
can leak information. For this reason, this plug also forbids
any GET JavaScript request that is not XHR (or AJAX).
Note that it is recommended to enable CSRFProtection whenever
a session is used, even for JSON requests. For example, Chrome
had a bug that allowed POST requests to be triggered with
arbitrary content-type, making JSON exploitable. More info:
https://bugs.chromium.org/p/chromium/issues/detail?id=490015
Finally, we recommend developers to invoke delete_csrf_token/0
every time after they log a user in, to avoid CSRF fixation
attacks.

  
    
  
  Token generation


This plug won't generate tokens automatically. Instead, tokens
will be generated only when required by calling get_csrf_token/0.
In case you are generating the token for certain specific URL,
you should use get_csrf_token_for/1 as that will avoid tokens
from being leaked to other applications.
Once a token is generated, it is cached in the process dictionary.
The CSRF token is usually generated inside forms which may be
isolated from Plug.Conn. Storing them in the process dictionary
allows them to be generated as a side-effect only when necessary,
becoming one of those rare situations where using the process
dictionary is useful.

  
    
  
  Cross-host protection


If you are sending data to a full URI, such as //subdomain.host.com/path
or //external.com/path, instead of a simple path such as /path, you may
want to consider using get_csrf_token_for/1, as that will encode the host
in the CSRF token. Once received, Plug will only consider the CSRF token to
be valid if the host encoded in the token is the same as the one in
conn.host.
Therefore, if you get a warning that the host does not match, it is either
because someone is attempting to steal CSRF tokens or because you have a
misconfigured host configuration.
For example, if you are running your application behind a proxy, the browser
will send a request to the proxy with www.example.com but the proxy will
request you using an internal IP. In such cases, it is common for proxies
to attach information such as "x-forwarded-host" that contains the original
host.
This may also happen on redirects. If you have a POST request to foo.example.com
that redirects to bar.example.com with status 307, the token will contain a
different host than the one in the request.
You can pass the :allow_hosts option to control any host that you may want
to allow. The values in :allow_hosts may either be a full host name or a
host suffix. For example: ["www.example.com", ".subdomain.example.com"]
will allow the exact host of "www.example.com" and any host that ends with
".subdomain.example.com".

  
    
  
  Options


	:session_key - the name of the key in session to store the token under
	:allow_hosts - a list with hosts to allow on cross-host tokens
	:with - should be one of :exception or :clear_session. Defaults to
:exception.	:exception -  for invalid requests, this plug will raise
Plug.CSRFProtection.InvalidCSRFTokenError.
	:clear_session -  for invalid requests, this plug will set an empty
session for only this request. Also any changes to the session during this
request will be ignored.




  
    
  
  Disabling


You may disable this plug by doing
Plug.Conn.put_private(conn, :plug_skip_csrf_protection, true). This was made
available for disabling Plug.CSRFProtection in tests and not for dynamically
skipping Plug.CSRFProtection in production code. If you want specific routes to
skip Plug.CSRFProtection, then use a different stack of plugs for that route that
does not include Plug.CSRFProtection.

  
    
  
  Examples


plug Plug.Session, ...
plug :fetch_session
plug Plug.CSRFProtection

A Plug to convert HEAD requests to GET requests.

  
    
  
  Examples


plug Plug.Head

A plug for logging basic request information in the format:
GET /index.html
Sent 200 in 572ms
To use it, just plug it into the desired module.
plug Plug.Logger, log: :debug

  
    
  
  Options


	:log - The log level at which this plug should log its request info.
Default is :info.
The list of supported levels
is available in the Logger documentation.

This plug overrides the request's POST method with the method defined in
the _method request parameter.
The POST method can be overridden only by these HTTP methods:
	PUT
	PATCH
	DELETE

This plug only replaces the request method if the _method request
parameter is a string. If the _method request parameter is not a string,
the request method is not changed.
Parse Body Parameters First
This plug expects the body parameters to be already fetched and
parsed. Those can be fetched with Plug.Parsers.

This plug doesn't accept any options.
To recap, here are all the conditions that the request must meet in order
for this plug to replace the :method field in the Plug.Conn:
	The conn's request :method must be POST.
	The conn's :body_params must have been fetched already (for example,
with Plug.Parsers).
	The conn's :body_params must have a _method field that is a string
and whose value is "PUT", "PATCH", or "DELETE" (case insensitive).


  
    
  
  Usage


# You'll need to fetch and parse parameters first, for example:
# plug Plug.Parsers, parsers: [:urlencoded, :multipart, :json]

plug Plug.MethodOverride

A plug for parsing the request body.
It invokes a list of :parsers, which are activated based on the
request content-type. Custom parsers are also supported by defining
a module that implements the behaviour defined by this module.
Once a connection goes through this plug, it will have :body_params
set to the map of params parsed by one of the parsers listed in
:parsers and :params set to the result of merging the :body_params
and :query_params. In case :query_params have not yet been parsed,
Plug.Conn.fetch_query_params/2 is automatically invoked.
This plug will raise Plug.Parsers.UnsupportedMediaTypeError by default if
the request cannot be parsed by any of the given types and the MIME type has
not been explicitly accepted with the :pass option.
Plug.Parsers.RequestTooLargeError will be raised if the request goes over
the given limit. The default length is 8MB and it can be customized by passing
the :length option to the Plug. :read_timeout and :read_length, as
described by Plug.Conn.read_body/2, are also supported.
Parsers may raise a Plug.Parsers.ParseError if the request has a malformed
body.
This plug only parses the body if the request method is one of the following:
	POST
	PUT
	PATCH
	DELETE

For requests with a different request method, this plug will only fetch the
query params.

  
    
  
  Options


	:parsers - a list of modules or atoms of built-in parsers to be
invoked for parsing. These modules need to implement the behaviour
outlined in this module.

	:pass - an optional list of MIME type strings that are allowed
to pass through. Any mime not handled by a parser and not explicitly
listed in :pass will raise UnsupportedMediaTypeError. For example:
	["*/*"] - never raises
	["text/html", "application/*"] - doesn't raise for those values
	[] - always raises (default)


	:query_string_length - the maximum allowed size for query strings

	:validate_utf8 - boolean that tells whether or not we want to
  validate that parsed binaries are utf8 strings.

	:body_reader - an optional replacement (or wrapper) for
Plug.Conn.read_body/2 to provide a function that gives access to the
raw body before it is parsed and discarded. It is in the standard format
of {Module, :function, [args]} (MFA) and defaults to
{Plug.Conn, :read_body, []}. Note that this option is not used by
Plug.Parsers.MULTIPART which relies instead on other functions defined
in Plug.Conn.


All other options given to this Plug are forwarded to the parsers.

  
    
  
  Examples


plug Plug.Parsers,
     parsers: [:urlencoded, :multipart],
     pass: ["text/*"]
Any other option given to Plug.Parsers is forwarded to the underlying
parsers. Therefore, you can use a JSON parser and pass the :json_decoder
option at the root:
plug Plug.Parsers,
     parsers: [:urlencoded, :json],
     json_decoder: Jason
Or directly to the parser itself:
plug Plug.Parsers,
     parsers: [:urlencoded, {:json, json_decoder: Jason}]
It is also possible to pass the :json_decoder as a {module, function, args} tuple,
useful for passing options to the JSON decoder:
plug Plug.Parsers,
     parsers: [:json],
     json_decoder: {Jason, :decode!, [[floats: :decimals]]}
A common set of shared options given to Plug.Parsers is :length,
:read_length and :read_timeout, which customizes the maximum
request length you want to accept. For example, to support file
uploads, you can do:
plug Plug.Parsers,
     parsers: [:urlencoded, :multipart],
     length: 20_000_000
However, the above will increase the maximum length of all request
types. If you want to increase the limit only for multipart requests
(which is typically the ones used for file uploads), you can do:
plug Plug.Parsers,
     parsers: [
       :urlencoded,
       {:multipart, length: 20_000_000} # Increase to 20MB max upload
     ]

  
    
  
  Built-in parsers


Plug ships with the following parsers:
	Plug.Parsers.URLENCODED - parses application/x-www-form-urlencoded
requests (can be used as :urlencoded as well in the :parsers option)
	Plug.Parsers.MULTIPART - parses multipart/form-data and
multipart/mixed requests (can be used as :multipart as well in the
:parsers option)
	Plug.Parsers.JSON - parses application/json requests with the given
:json_decoder (can be used as :json as well in the :parsers option)


  
    
  
  File handling


If a file is uploaded via any of the parsers, Plug will
stream the uploaded contents to a file in a temporary directory in order to
avoid loading the whole file into memory. For such, the :plug application
needs to be started in order for file uploads to work. More details on how the
uploaded file is handled can be found in the documentation for Plug.Upload.
When a file is uploaded, the request parameter that identifies that file will
be a Plug.Upload struct with information about the uploaded file (e.g.
filename and content type) and about where the file is stored.
The temporary directory where files are streamed to can be customized by
setting the PLUG_TMPDIR environment variable on the host system. If
PLUG_TMPDIR isn't set, Plug will look at some environment
variables which usually hold the value of the system's temporary directory
(like TMPDIR or TMP). If no value is found in any of those variables,
/tmp is used as a default.

  
    
  
  Custom body reader


Sometimes you may want to customize how a parser reads the body from the
connection. For example, you may want to cache the body to perform verification
later, such as HTTP Signature Verification. This can be achieved with a custom
body reader that would read the body and store it in the connection, such as:
defmodule CacheBodyReader do
  def read_body(conn, opts) do
    {:ok, body, conn} = Plug.Conn.read_body(conn, opts)
    conn = update_in(conn.assigns[:raw_body], &[body | (&1 || [])])
    {:ok, body, conn}
  end
end
which could then be set as:
plug Plug.Parsers,
  parsers: [:urlencoded, :json],
  pass: ["text/*"],
  body_reader: {CacheBodyReader, :read_body, []},
  json_decoder: Jason

Plug.RequestId 
    



      
A plug for generating a unique request ID for each request.
The generated request ID will be in the format:
uq8hs30oafhj5vve8ji5pmp7mtopc08f
If a request ID already exists in a configured HTTP request header (see options below),
then this plug will use that value, assuming it is between 20 and 200 characters.
If such header is not present, this plug will generate a new request ID.
The request ID is added to the Logger metadata as :request_id, and to the
response as the configured HTTP response header (see options below). To see the
request ID in your log output, configure your logger backends to include the :request_id
metadata. For example:
config :logger, :console, metadata: [:request_id]
We recommend to include this metadata configuration in your production
configuration file.
Programmatic access to the request ID
To access the request ID programmatically, use the :assign_as option (see below)
to assign the request ID to a key in conn.assigns, and then fetch it from there.


  
    
  
  Usage


To use this plug, just plug it into the desired module:
plug Plug.RequestId

  
    
  
  Options


	:http_header - The name of the HTTP request header to check for
existing request IDs. This is also the HTTP response header that will be
set with the request id. Default value is "x-request-id".
plug Plug.RequestId, http_header: "custom-request-id"

	:assign_as - The name of the key that will be used to store the
discovered or generated request id in conn.assigns. If not provided,
the request id will not be stored. Available since v1.16.0.
plug Plug.RequestId, assign_as: :plug_request_id

A plug to rewrite the request's host/port/protocol from x-forwarded-* headers.
If your Plug application is behind a proxy that handles HTTPS, you may
need to tell Plug to parse the proper protocol from the x-forwarded-*
header.
plug Plug.RewriteOn, [:x_forwarded_host, :x_forwarded_port, :x_forwarded_proto]
The supported values are:
	:x_forwarded_for - to override the remote ip based on the "x-forwarded-for" header
	:x_forwarded_host - to override the host based on the "x-forwarded-host" header
	:x_forwarded_port - to override the port based on the "x-forwarded-port" header
	:x_forwarded_proto - to override the protocol based on the "x-forwarded-proto" header

A tuple representing a Module-Function-Args can also be given as argument
instead of a list.
Since rewriting the scheme based on x-forwarded-* headers can open up
security vulnerabilities, only use this plug if:
	your app is behind a proxy
	your proxy strips the given x-forwarded-* headers from all incoming requests
	your proxy sets the x-forwarded-* headers and sends it to Plug

A plug to force SSL connections and enable HSTS.
If the scheme of a request is https, it'll add a strict-transport-security
header to enable HTTP Strict Transport Security by default.
Otherwise, the request will be redirected to a corresponding location
with the https scheme by setting the location header of the response.
The status code will be 301 if the method of conn is GET or HEAD,
or 307 in other situations.
Besides being a Plug, this module also provides conveniences for configuring
SSL. See configure/1.

  
    
  
  x-forwarded-*


If your Plug application is behind a proxy that handles HTTPS, you may
need to tell Plug to parse the proper protocol from the x-forwarded-*
header. This can be done using the :rewrite_on option:
plug Plug.SSL, rewrite_on: [:x_forwarded_host, :x_forwarded_port, :x_forwarded_proto]
For further details refer to Plug.RewriteOn.

  
    
  
  Plug Options


	:rewrite_on - rewrites the given connection information based on the given headers
	:hsts - a boolean on enabling HSTS or not, defaults to true
	:expires - seconds to expires for HSTS, defaults to 31_536_000 (1 year)
	:preload - a boolean to request inclusion on the HSTS preload list
(for full set of required flags, see: Chromium HSTS submission site),
defaults to false
	:subdomains - a boolean on including subdomains or not in HSTS,
defaults to false
	:exclude - exclude the given hosts from redirecting to the https
scheme. Defaults to ["localhost"]. It may be set to a list of binaries
or a tuple {module, function, args}.
	:host - a new host to redirect to if the request's scheme is http,
defaults to conn.host. It may be set to a binary or a tuple
{module, function, args} that will be invoked on demand
	:log - The log level at which this plug should log its request info.
Default is :info. Can be false to disable logging.


  
    
  
  Port


It is not possible to directly configure the port in Plug.SSL because
HSTS expects the port to be 443 for SSL. If you are not using HSTS and
want to redirect to HTTPS on another port, you can sneak it alongside
the host, for example: host: "example.com:443".

  
    
  
  Excluded hosts tuple


Tuple {module, function, args} can be passed to be invoked each time
the plug is checking whether to redirect host. Provided function needs
to receive at least one argument (host).
For example, you may define it as:
plug Plug.SSL,
  rewrite_on: [:x_forwarded_proto],
  exclude: {__MODULE__, :excluded_host?, []}
where:
def excluded_host?(host) do
  # Custom logic
end

A plug to handle session cookies and session stores.
The session is accessed via functions on Plug.Conn. Cookies and
session have to be fetched with Plug.Conn.fetch_session/1 before the
session can be accessed.
The session is also lazy. Once configured, a cookie header with the
session will only be sent to the client if something is written to the
session in the first place.
When using Plug.Session, also consider using Plug.CSRFProtection
to avoid Cross Site Request Forgery attacks.

  
    
  
  Session stores


See Plug.Session.Store for the specification session stores are required to
implement.
Plug ships with the following session stores:
	Plug.Session.ETS
	Plug.Session.COOKIE


  
    
  
  Options


	:store - session store module (required);
	:key - session cookie key (required);
	:domain - see Plug.Conn.put_resp_cookie/4;
	:max_age - see Plug.Conn.put_resp_cookie/4;
	:path - see Plug.Conn.put_resp_cookie/4;
	:secure - see Plug.Conn.put_resp_cookie/4;
	:http_only - see Plug.Conn.put_resp_cookie/4;
	:same_site - see Plug.Conn.put_resp_cookie/4;
	:extra - see Plug.Conn.put_resp_cookie/4;

Additional options can be given to the session store, see the store's
documentation for the options it accepts.

  
    
  
  Examples


plug Plug.Session, store: :ets, key: "_my_app_session", table: :session

Plug.Static 
    



      
A plug for serving static assets.
It requires two options:
	:at - the request path to reach for static assets.
It must be a string.

	:from - the file system path to read static assets from.
It can be either: a string containing a file system path, an
atom representing the application name (where assets will
be served from priv/static), a tuple containing the
application name and the directory to serve assets from (besides
priv/static), or an MFA tuple.


The preferred form is to use :from with an atom or tuple, since
it will make your application independent from the starting directory.
For example, if you pass:
plug Plug.Static, from: "priv/app/path"
Plug.Static will be unable to serve assets if you build releases
or if you change the current directory. Instead do:
plug Plug.Static, from: {:app_name, "priv/app/path"}
If a static asset cannot be found, Plug.Static simply forwards
the connection to the rest of the pipeline.

  
    
  
  Cache mechanisms


Plug.Static uses etags for HTTP caching. This means browsers/clients
should cache assets on the first request and validate the cache on
following requests, not downloading the static asset once again if it
has not changed. The cache-control for etags is specified by the
cache_control_for_etags option and defaults to "public".
However, Plug.Static also supports direct cache control by using
versioned query strings. If the request query string starts with
"?vsn=", Plug.Static assumes the application is versioning assets
and does not set the ETag header, meaning the cache behaviour will
be specified solely by the cache_control_for_vsn_requests config,
which defaults to "public, max-age=31536000".

  
    
  
  Options


	:encodings - list of 2-ary tuples where first value is value of
the Accept-Encoding header and second is extension of the file to
be served if given encoding is accepted by client. Entries will be tested
in order in list, so entries higher in list will be preferred. Defaults
to: [].
In addition to setting this value directly it supports 2 additional
options for compatibility reasons:
	:brotli - will append {"br", ".br"} to the encodings list.
	:gzip - will append {"gzip", ".gz"} to the encodings list.

Additional options will be added in the above order (Brotli takes
preference over Gzip) to reflect older behaviour which was set due
to fact that Brotli in general provides better compression ratio than
Gzip.

	:cache_control_for_etags - sets the cache header for requests
that use etags. Defaults to "public".

	:etag_generation - specify a {module, function, args} to be used
to generate   an etag. The path of the resource will be passed to
the function, as well as the args. If this option is not supplied,
etags will be generated based off of file size and modification time.
Note it is recommended for the etag value to be quoted,
which Plug won't do automatically.

	:cache_control_for_vsn_requests - sets the cache header for
requests starting with "?vsn=" in the query string. Defaults to
"public, max-age=31536000".

	:only - filters which requests to serve. This is useful to avoid
file system access on every request when this plug is mounted
at "/". For example, if only: ["images", "favicon.ico"] is
specified, only files in the "images" directory and the
"favicon.ico" file will be served by Plug.Static.
Note that Plug.Static matches these filters against request
uri and not against the filesystem. When requesting
a file with name containing non-ascii or special characters,
you should use urlencoded form. For example, you should write
only: ["file%20name"] instead of only: ["file name"].
Defaults to nil (no filtering).

	:only_matching - a relaxed version of :only that will
serve any request as long as one of the given values matches the
given path. For example, only_matching: ["images", "favicon"]
will match any request that starts at "images" or "favicon",
be it "/images/foo.png", "/images-high/foo.png", "/favicon.ico"
or "/favicon-high.ico". Such matches are useful when serving
digested files at the root. Defaults to nil (no filtering).

	:headers - other headers to be set when serving static assets. Specify either
an enum of key-value pairs or a {module, function, args} to return an enum. The
conn will be passed to the function, as well as the args.

	:content_types - controls custom MIME type mapping.
It can be a map with filename as key and content type as value to override
the default type for matching filenames. For example:
content_types: %{"apple-app-site-association" => "application/json"}.
Alternatively, it can be the value false to opt out of setting the header at all. The latter
can be used to set the header based on custom logic before calling this plug.
Defaults to an empty map %{}.



  
    
  
  Examples


This plug can be mounted in a Plug.Builder pipeline as follows:
defmodule MyPlug do
  use Plug.Builder

  plug Plug.Static,
    at: "/public",
    from: :my_app,
    only: ~w(images robots.txt)
  plug :not_found

  def not_found(conn, _) do
    send_resp(conn, 404, "not found")
  end
end

A plug to instrument the pipeline with :telemetry events.
When plugged, the event prefix is a required option:
plug Plug.Telemetry, event_prefix: [:my, :plug]
In the example above, two events will be emitted:
	[:my, :plug, :start] - emitted when the plug is invoked.
The event carries the system_time as measurement. The metadata
is the whole Plug.Conn under the :conn key and any leftover
options given to the plug under :options.

	[:my, :plug, :stop] - emitted right before the response is sent.
The event carries a single measurement, :duration,  which is the
monotonic time difference between the stop and start events.
It has the same metadata as the start event, except the connection
has been updated.


Note this plug measures the time between its invocation until a response
is sent. The :stop event is not guaranteed to be emitted in all error
cases, so this Plug cannot be used as a Telemetry span.

  
    
  
  Time unit


The :duration measurements are presented in the :native time unit.
You can read more about it in the docs for System.convert_time_unit/3.

  
    
  
  Example


defmodule InstrumentedPlug do
  use Plug.Router

  plug :match
  plug Plug.Telemetry, event_prefix: [:my, :plug]
  plug Plug.Parsers, parsers: [:urlencoded, :multipart]
  plug :dispatch

  get "/" do
    send_resp(conn, 200, "Hello, world!")
  end
end
In this example, the stop event's duration includes the time
it takes to parse the request, dispatch it to the correct handler,
and execute the handler. The events are not emitted for requests
not matching any handlers, since the plug is placed after the match plug.

A module (not a plug) for debugging in development.
This module is commonly used within a Plug.Builder or a Plug.Router
and it wraps the call/2 function.
Notice Plug.Debugger does not catch errors, as errors should still
propagate so that the Elixir process finishes with the proper reason.
This module does not perform any logging either, as all logging is done
by the web server handler.
Note: If this module is used with Plug.ErrorHandler, only one of
them will effectively handle errors. For this reason, it is recommended
that Plug.Debugger is used before Plug.ErrorHandler and only in
particular environments, like :dev.
In case of an error, the rendered page drops the content-security-policy
header before rendering the error to ensure that the error is displayed
correctly.

  
    
  
  Examples


defmodule MyApp do
  use Plug.Builder

  if Mix.env == :dev do
    use Plug.Debugger, otp_app: :my_app
  end

  plug :boom

  def boom(conn, _) do
    # Error raised here will be caught and displayed in a debug page
    # complete with a stacktrace and other helpful info.
    raise "oops"
  end
end

  
    
  
  Options


	:otp_app - the OTP application that is using Plug. This option is used
to filter stacktraces that belong only to the given application.
	:style - custom styles (see below)
	:banner - the optional MFA ({module, function, args}) which receives
exception details and returns banner contents to appear at the top of
the page. May be any string, including markup.


  
    
  
  Custom styles


You may pass a :style option to customize the look of the HTML page.
use Plug.Debugger, style:
  [primary: "#c0392b", logo: "data:image/png;base64,..."]
The following keys are available:
	:primary - primary color
	:accent - accent color
	:logo - logo URI, or nil to disable

The :logo is preferred to be a base64-encoded data URI so not to make any
external requests, though external URLs (eg, https://...) are supported.

  
    
  
  Custom Banners


You may pass an MFA ({module, function, args}) to be invoked when an
error is rendered which provides a custom banner at the top of the
debugger page. The function receives the following arguments, with the
passed args concatenated at the end:
[conn, status, kind, reason, stacktrace]
For example, the following :banner option:
use Plug.Debugger, banner: {MyModule, :debug_banner, []}
would invoke the function:
MyModule.debug_banner(conn, status, kind, reason, stacktrace)

  
    
  
  Links to the text editor


If a PLUG_EDITOR environment variable is set, Plug.Debugger will
use it to generate links to your text editor. The variable should be
set with __FILE__ and __LINE__ placeholders which will be correctly
replaced. For example (with the TextMate editor):
txmt://open/?url=file://__FILE__&line=__LINE__
Or, using Visual Studio Code:
vscode://file/__FILE__:__LINE__

A module to be used in your existing plugs in order to provide
error handling.
defmodule AppRouter do
  use Plug.Router
  use Plug.ErrorHandler

  plug :match
  plug :dispatch

  get "/hello" do
    send_resp(conn, 200, "world")
  end

  @impl Plug.ErrorHandler
  def handle_errors(conn, %{kind: _kind, reason: _reason, stack: _stack}) do
    send_resp(conn, conn.status, "Something went wrong")
  end
end
Once this module is used, a callback named handle_errors/2 should
be defined in your plug. This callback will receive the connection
already updated with a proper status code for the given exception.
The second argument is a map containing:
	the exception kind (:throw, :error or :exit),
	the reason (an exception for errors or a term for others)
	the stacktrace

After the callback is invoked, the error is re-raised.
It is advised to do as little work as possible when handling errors
and avoid accessing data like parameters and session, as the parsing
of those is what could have led the error to trigger in the first place.
Also notice that these pages are going to be shown in production. If
you are looking for error handling to help during development, consider
using Plug.Debugger.
Note: If this module is used with Plug.Debugger, it must be used
after Plug.Debugger.

Plug.Exception protocol
    



      
A protocol that extends exceptions to be status-code aware.
By default, it looks for an implementation of the protocol,
otherwise checks if the exception has the :plug_status field
or simply returns 500.

Specification of the connection adapter API implemented by webservers.

Conveniences for encoding and decoding cookies.

Conveniences for decoding and encoding URL-encoded queries.
Plug allows developers to build query strings that map to
Elixir structures in order to make manipulation of such structures
easier on the server side. Here are some examples:
iex> decode("foo=bar")["foo"]
"bar"
If a value is given more than once, the last value takes precedence:
iex> decode("foo=bar&foo=baz")["foo"]
"baz"
Nested structures can be created via [key]:
iex> decode("foo[bar]=baz")["foo"]["bar"]
"baz"
Lists are created with []:
iex> decode("foo[]=bar&foo[]=baz")["foo"]
["bar", "baz"]
Nesting inside lists
Nesting inside lists is ambiguous and unspecified behavio